C3M2

Plotting in Cylindrical and Spherical Coordinates

There are two basic 3d-plotters which will be introduced in this section. The first is ‘ `cylinderplot`’ and the second is ‘ `sphereplot`’ which use cylindrical coordinates and spherical coordinates as their bases respectively. The syntax for each may involve a standard form or a parametric form. While we may use each form, the parametric is the most useful, especially when the surface to be plotted is vertical. For now we will concentrate on `cylinderplot`. When the surface is defined by $z = f(r, \theta)$, for r between $g(\theta)$ and $h(\theta)$, and θ between a and b we would use

```
> cylinderplot(f(r,t),r=g(t)..h(t),t=a..b,color=green);
```

to obtain our surface in green. The parametric form uses square braces and coordinates $[r, \theta, z]$ so the same result occurs from

```
> cylinderplot([r,t,f(r,t)],r=g(t)..h(t),t=a..b,color=green);
```

It is easy to see that we have used t instead of θ here.

Before we begin it is helpful to understand what results when one of the variables is held constant in cylindrical coordinates. With coordinates $[r, \theta, z]$, a vertical cylinder is produced when r is held constant and the other variables have constant bounds. Then, when θ is constant a rectangular portion of a vertical plane containing the z-axis is determined. Last, if z is constant, a portion of an annular ring that is horizontal results. Think of a piece of pie with a bite taken out of the part closest to the center.

Example 1: Use cylindrical coordinates to sketch the quarter of the vertical cylinder $x^2 + y^2 \leq 1$ bounded above by the sphere $x^2 + y^2 + z^2 = 4$ that lies in the first octant. Assume that the package ‘plots’ has been included above in the worksheet.

The top is given by:

```
> roof:=cylinderplot([r,t,sqrt(4-r^2),r=0..1,t=0..Pi/2,color=blue):
```

The bottom comes from:

```
> floor1:=cylinderplot([r,t,0],r=0..1,t=0..Pi/2,color=green):
```

The curved side of the cylinder has its radius held constant at 1, while the angle and height change.

```
> cwall:=cylinderplot([1,t,z],t=0..Pi/2,z=0..sqrt(3),color=red):
```

The side that lies in the plane $x = 0$ corresponds to having $\theta = \pi/2$, so:

```
> side1:=cylinderplot([r,Pi/2,z],z=0..sqrt(4-r^2),r=0..1,color=plum):
```

Note how z varied between two curves, $z = 0$ and $z = 4 - r^2$ here and when $y = 0$. The other vertical wall occurs when $\theta = 0$ and:

```
> side2:=cylinderplot([r,0,z],z=0..sqrt(4-r^2),r=0..1,color=cyan):
```

Putting them all together,

```
> display(floor1,roof,cwall,side1,side2);
```



The reader is encouraged to enter the commands above and observe the results. Rotate the solid so that you can see each of the surfaces. Save this example because we will use it later and will observe how to find its volume using a triple integral.

Example 2: Use cylindrical coordinates to plot the solid that lies inside the sphere $x^2+y^2+z^2=9$, above the xy-plane, below the upper portion of the cone $x^2+y^2=z^2$, and excludes the first octant.

We begin by noting that in cylindrical coordinates the sphere is $r^2+z^2=9$ and the cone is $r=z$. By substitution, we see that these surfaces intersect when $r=z=3/\sqrt{2}$. We begin with the cone:

```
> cone2:=cylinderplot([r,t,r],r=0..3/sqrt(2), t=Pi/2..2*Pi,color=blue):
```

Now for the spherical surface:

```
> sph:=cylinderplot([r,t,sqrt(9-r^2)],r=3/sqrt(2)..3,t=Pi/2..2*Pi,color=red):
```

Pay close attention to how we plot the sides. If we tried to let z vary between two curves, what would happen? The angle θ is a constant. We must let r vary between two curves. In a parametric plot of a surface it normally happens that one variable varies between two curves, while the other lies between two values. Begin to think: "Curve-to-curve, point-to-point".

```
> side1:=cylinderplot([r,0,z],r=z..sqrt(9-z^2),z=0..3/sqrt(2),color=cyan):
```

On the other side $\theta=\pi/2$:

```
> side2:=cylinderplot(r,Pi/2,z],r=z..sqrt(9-z^2),z=0..3/sqrt(2),color=cyan):
```

The bottom remains and then we display them all together, with the resulting figure on the left.

```
> floor2:=cylinderplot([r,t,0],r=0..3, t=Pi/2..2*Pi,color=green):
> display(floor2,cone2,sph,side1,side2);
```





Now we turn our attention to spherical coordinates where a point is identified by $[\rho, \theta, \phi]$. The command ' `sphereplot(f(theta,phi),theta=g(phi)..h(phi),phi=a..b);`' is used when $\rho = f(\theta, \phi)$. It is a little easier to use `sphereplot` in the parametric form where one of the variables is a function of the other two. For example, to obtain all of a magenta sphere of radius 2 centered at the origin,

```
> sphereplot([2,theta,phi],theta=0..2*Pi,phi=0..Pi,color=magenta);
```

The value of ρ is the distance to the point from the origin. The angle θ is the same as that used in cylindrical coordinates. Remember, the angle ϕ is measured down from the z-axis to the point. We will begin with the example that we just completed, but will use spherical coordinates. All the variables will lie between constants. Please think about this last statement.

First, let's consider the consequences of holding each of the variables constant in spherical coordinates. In a system with $[\rho, \theta, \phi]$, a portion of a sphere of fixed radius ρ occurs when ρ is a constant. Second, a part of a vertical plane containing the usual z-axis is determined. It would look like a piece of pie on its side. Last, ϕ constant determines a cone with the center axis along the usual z-axis. If $0 < \phi \leq \pi/2$, the cone opens upward and will hold water, while if $\pi/2 \leq \phi < \pi$, it opens downward, like a traffic cone/pylon.

Example 3: Use spherical coordinates to plot the solid that lies inside the sphere $x^2 + y^2 + z^2 = 9$, above the xy-plane, below the upper portion of the cone $x^2 + y^2 = z^2$, and excludes the first octant.

The reader is encouraged to analyze each line and compare carefully with the results of Example 2. This figure is on the right above.

```
> with(plots):
> sph3:=sphereplot([3,theta,phi],theta=Pi/2..2*Pi,phi=Pi/4..Pi/2,color=red):
> cone3:=sphereplot([rho,theta,Pi/4], theta=Pi/2..2*Pi,rho=0..3,color=blue):
> side3:=sphereplot([rho,Pi/2,phi],rho=0..3,phi=Pi/4..Pi/2,color=cyan):
> side4:=sphereplot([rho,0,phi],rho=0..3,phi=Pi/4..Pi/2,color=cyan):
> bottom3:=sphereplot([rho,theta,Pi/2], theta=Pi/2..2*Pi,rho=0..3,color=green):
> display(sph3,cone3,side3,side4,bottom3);
```

The last example is a little more challenging. Recall that in polar coordinates we learned that a vertical line, $x = a$, has the polar equation $r = a \sec\theta$. A horizontal line $y = b$ translates to $r = b\csc\theta$. You are requested to insert the lines below in a worksheet and determine their consequences. In each case ask yourself where $\theta = 0$ and $\theta = \pi/2$ are located in the polar system and where $\phi = 0$ and $\phi = \pi/2$ are located in the spherical system.

```
> sphereplot([2*sec(phi),theta,phi],theta=0..2*Pi,phi=0..Pi/4,color=cyan);
> sphereplot([2*csc(phi),theta,phi],theta=0..2*Pi,phi=Pi/4..3*Pi/4,color=red);
```

WARNING! If you click on the secant plot and click on an axis sphere, then the system may hang up!

Example 4 Plot the solid that lies inside the sphere $x^2 + y^2 + z^2 = 4$ and is **outside** the cylinder $x^2 + y^2 = 1$.

A three-quarter view of this solid results from:

```
> sph4:=sphereplot([2,theta,phi],theta=Pi/2..2*Pi,phi=Pi/6..5*Pi/6):
> cyl4:=sphereplot([csc(phi),theta,phi],theta=Pi/2..2*Pi,phi=Pi/6..5*Pi/6):
> side5:=sphereplot([rho,0,phi],rho=csc(phi)..2,phi=Pi/6..5*Pi/6):
```

```
> side6:=sphereplot([rho,Pi/2,phi],rho=csc(phi)..2,phi=Pi/6..5*Pi/6):
> display(sph4,cyl4,side5,side6);
```



C3M2 Problems Use cylinderplot and/or sphereplot to plot the solids listed:

1. Q is bounded above by $z = 4$, below by $z = x^2 + y^2$, and on the side by $x^2 + y^2 = 1$.

2. R is a "slice of cheese". It is bounded above by $z = 2$, below by $z = 0$, on the sides by $x = 0$ and $y = x$, and on the outside by $x^2 + y^2 = 4$.

3. S is inside the the sphere $x^2 + y^2 + z^2 = 9$ and above the cone $3z^2 = x^2 + y^2$.

4. T is the portion of the upper hemisphere $x^2 + y^2 + z^2 = 4$ with $z \geq 0$ that lies between the planes $y = x$ and $y = \sqrt{3}x$ in the first octant.